
 Fastest scorer in football in Sweden

Since 1959 a daily local newspaper has recorded the fastest scorer in football (soccer) in Sweden at the start of the season. Every player (or at least position in the team in the case of one player taking over after another one) plays his first match for the season. (We disregard the possibility of a team losing a player because of misbehaviour in which case the player is not replaced.) 

Thus we have a number of matches that are played in the first round. If we consider the process of scoring goals as a process, the most natural model, or at least the first most natural model, is the Poisson process and we in such case conclude that the times between goals are distributed accor​ding to an exponential distribution. (See the documents A course in statistics, A collection of dia​grams or Exercises with computer support for a discussion of the distri​bution.)

A theoretical discussion.  Before looking at the real football data we consider some theoretical aspects of the exponential distribution. The time to the first goal depends not only of the intensity of scoring, which of course is different in different sports or different leagues. It also depends on the number of matches considered. The more matches that are included in this competition the shorter the time is to the first goal at a given intensity. An analogy to this is that the strength of a chain is lower the longer the chain is, as the chance of (randomly) including a weak link increases. We show this fact by some simulations:

random 1000 c1-c20;          # Simulates exponential time data in hours.
expo 0.5.                    # Each row is a sample of 20 matches.

                             # The average is approx 0.5 h between goals.

                             # 1000 ’start of season’ are simulated.

rmin c1-c20 c501             # For each ’start of season’ the (row) minimum

                             # of 20 matches is stored in c501.

random 1000 c1-c50;          # The same as above but for 50 matches.
expo 0.5.

rmin c1-c50 c502             # Result in c502.

random 1000 c1-c100;         # The same as above but for 100 matches.
expo 0.5.

rmin c1-c100 c503            # Result in c503.

describe c501-c503           # Describes c501-c503.

stack c501-c503 c504         # Stacks all data in column c504 for plotting.

set c505                     # Stores 1000 ’1’, 1000 ’2’ and 1000 ’3’ in
(1 2 3)1000                  # c505 to be used in following plot.
end

plot c504*c505               # Plots ’minimum time’ against group size.

It is very obvious in the diagram that the larger the group considered, the smaller the time to first score becomes. It is also very clear that time has a lower variation when the group size increases. This is a fact of the exponential distribution. Before we proof that the data for the different groups are exponen​tially distributed, we make a test using the statistical software:

copy c504 c511;              # Copies all ’group 1’ data to c511.
use c505 1.

copy c504 c512;              # Copies all ’group 2’ data to c512.
use c505 2.

copy c504 c513;              # Copies all ’group 3’ data to c513.
use c505 3.

Use the menus [Graph]>[Probability Plots…] or use the macro %Disttest to see if the columns c511 – c512 can be considered exponentially distributed.

A statistical reasoning.  Let us designate the time variables (i.e. time to first score) as X1, X2 … Xn. We then let Y be the following:

In words it means that if we have the result from a number of variables (n), we let Y be the minimum of these. Now we define the following event (” The minimum time is larger than y”) 

, and in order for this to be true we must have 
[image: image3.wmf])

(

1

y

X

³

 and 
[image: image4.wmf])

(

2

y

X

³

 and 
[image: image5.wmf])

(

3

y

X

³

 and 
[image: image6.wmf])

(

4

y

X

³

 and 
[image: image7.wmf])

(

5

y

X

³

 and 
[image: image8.wmf])

(

6

y

X

³

 and … and 
[image: image9.wmf])

(

y

X

n

³

.

Now we have the following fact for the exponential distribution (its so-called distribution function):

If all the variables are independent and have the same distribution we can simplify this to the following expression:

We can use the distribution function of the exponential distribution again and get the following:

This leads us to the fact that the time to the fastest scoring is exponentially distributed with the following distribution:

Test of the simulated data.  In column c511 – c513 we have the times to the fastest scoring for groups of 20, 50 and 100 matches at the start of the season. Each column contains the result of 1000 such ’start of season’. We should thus be able to estimate the parameter in each such process.

The theoretical parameter of the exponential distribution is known, namely n where  = 2 (see the simulation above. NB that the software Minitab uses the inverted value and thus separates itself from the usual convention found in the statistical literature). We show this as a simple table:

	Column
	Group size n
	Theoretical parameter n

	c511
	20

	c512
	50

	c513
	100

How can we check this against the data? We use the following facts of the exponential distribution:

In order to compare this with the data we can use the following commands:

let k1 = 1/mean(c511)        # Stores the estimated parameter in k1.

let k2 = 1/mean(c512)        # Stores the estimated parameter in k2.

let k3 = 1/mean(c513)        # Stores the estimated parameter in k3.

print k1-k3                  # Prints the three parameter estimates.

Conclusion.  Comparing the result of the print-out and the table above we most likely accept the thought that the smallest value of a number of exponentially distributed variables is again an expo​nentially distributed variable. If the original variables have the parameter  then the minimum has the parameter n.

Back to the football data.  Below we write the times to the first goal scored for a number of seasons since 1959. First we enter the data into some columns of Minitab:

set c1            # The year of the start of the season

    1959    1960    1961    1962    1963    1964    1965    1966    1967

    1968    1969    1970    1971    1972    1973    1974    1975    1976

    1977    1978    1979    1980    1981    1982    1983    1984    1985

    1986    1987    1988    1989    1990    1991    1992    1993    1994

    1995    1996    1997    1998

end

set c2            # The number of minutes from start of match

      7      6      5      0      3      1      1      3      0      5

      2     11      2      1      4      9      8      6      6      9

     11      4     21      3     32      3      7      0      4      3

      1     17      3      7     12     10     11     19      8     14

end

set c3            # The number of seconds

     16     53     14     26     40     16     28     35     25     42

     31     23     31     46      7     58     33      8     47     17

     27     23     32     34     32     20     50     34     17     35

     28     38     51     27     56     23      3     27     25     10

end

let c4 = c2 + c3/60           # Creating the times in decimal minutes.

We can now test this data and see whether we can accept it as being exponentially distributed. We do that by using the macro %Disttest which tests some different continuous variables. We can also use the menus [Graph]>[Probability Plot…] and use the exponential distribution. Neither test will reject the hypothesis that the time to the first scoring is exponentially distributed.

Another view.  From the statistical theory we know that the mean and the standard deviation of an ex​po​nential distribution is the same. But in the data above we got the results 7.47 and 6.57 from the football data. Is this a large deviation? We investigate and illuminate this question by some simu​lations via 1000 samples of 40 observations from a exponential distribution with mean value 7.47. 

For each sample we calculate the average and the standard deviation and plot these to values against each other. Finally we compare the values that we really got from the football data above. 

random 1000 c1-c40;              # Creates 1000 rows of data in c1-c40,
expo 7.47.                       # exponentially distributed.

rmean c1-c40 c51                 # The row-average, result in c51.

rstand c1-c40 c52                # The row-stand dev, result in c52.

Before plotting we also calculate the mean and standard deviation of all the values. This result will be shown as a blue dot in the plot below. We also include a red cross for the parameter used for the simulation (i.e. 7.47).

stack c1-c40 c41                 # All data is stored in c1.

let k4 = mean(c41)               # Total mean of all simulated data.

let k5 = stand(c41)              # Total stand dev of all simulated data.

The following commands performs a plot of the results above:

plot c51*c52;                    # Plots mean against the stand dev.
Title '1000 simulated samples from Exp[7.47], n = 40';
tsize 1.1;                       # Textsize.
Title '-  Red dot: average and standard deviation according to the football data  -';
tsize 0.8;                       # Textsize.
tcolor 2;                        # Gives red text.
Title '- Blue dot: average and standard deviation according to the simulated data –';
tsize 0.8;
tcolor 4;
marker 6.57 7.47;                # A marker according to football-data
color 2;                         # with red colour
size 2;                          # and of size 2.
marker 7.47 7.47;                # A marker according to the parameter
color 2;                         # used for simulation, red color
size 2;                          # and of size 2
type 6;                          # and type 6 (filled dot).
axlabel 1 'Standard deviation';  # Puts a label on the X-axis.
 tsize 0.9;                      # Size of text.
axlabel 2 'Average value';       # Puts a label on the Y-axis.
tsize 0.9;
data;                            # These two lines removes a frame around
etype 0;                         # the data area.

marker k5 k4;                    # A marker showing the total mean and
color 4;                         # standard deviation simulated data.
size 2;

type 6.

It is rather obvious that the time to the first scoring at the start of the football season can be re​garded as coming from an exponential distribution. In the plot above the average and the standard deviation from the football data (the red dot) is not in any way unusual, especially not compared to what we can expect from the randomness alone, modelled by the simulation.  ■

